

Carolyn Hughes
**PRIMEIROS PASSOS
COM O SEU
SINCLAIR (TKs)**

Este livro é dedicado a você.

Você que está prestes a entrar no mundo fascinante da computação.

Ele vai acompanhá-lo, passo a passo, do primeiro toque no teclado ao desenvolvimento de seus próprios programas.

- Descubra os segredos do seu TK (Sinclair)
- Aprenda a falar em Basic com seu equipamento
- Exercite suas habilidades com programas especiais
- Movimente as figuras pela tela
- Divirta-se com seis excitantes programas de jogos

Já é tempo de saber tudo que seu micro pode fazer — com sua ajuda, naturalmente.

ISBN 85-7001-217-9

(Edição original: ISBN 0-00-692240-6 An Armada Original, London)

PRIMEIROS PASSOS COM O SEU SINCLAIR (TKs)

Carolyn Hughes
**PRIMEIROS PASSOS
COM O SEU
SINCLAIR (TKs)**



Carolyn Hughes
**PRIMEIROS PASSOS
COM O SEU
SINCLAIR (TKs)**

Ilustrações
Bryan Reading

Tradução
Mário del Guercio Dolacio Mendes

EDITORA CAMPUS LTDA.
Rio de Janeiro

Do original:
First Steps With Your Spectrum
Copyright © 1983 by Carolyn Hughes.
Illustrations © Armada Paperbacks 1983

© 1985, Editora Campus Ltda.

Todos os direitos para a língua portuguesa reservados e protegidos pela Lei 5988 de 14/12/1973.

Nenhuma parte deste livro poderá ser reproduzida ou transmitida sejam quais forem os meios empregados: eletrônicos, mecânicos, fotográficos, gravação ou quaisquer outros.

Todo o esforço foi feito para fornecer a mais completa e adequada informação. Contudo a editora não assume responsabilidades pelo uso da mesma.

A Editora Campus não é filiada a nenhum fabricante de sistemas computacionais.

Capa
Otavio Studart

Diagramação, composição, paginação e revisão
Editora Campus Ltda.

Rua Barão de Itapagipe 55 Rio Comprido
Tel.: (021) 284 8443

20261 Rio de Janeiro RJ Brasil
Endereço telegráfico: CAMPUSRIO

ISBN 85-7001-217-9

(Edição original: ISBN 0-00-692240-6 An Armada Original, London.)

Ficha Catalográfica
CIP-Brasil. Catalogação-na-fonte
Sindicato Nacional dos Editores de Livros, RJ.

H888p Hughes, Carolyn.
Primeiros passos com o seu Sinclair TKs / Carolyn Hughes ; ilustrações de Bryan Beading ; tradução de Mário del Guercio Dolacio Mendes. — Rio de Janeiro : Campus, 1985.

Tradução de: First steps with your Spectrum.
ISBN 85-7001-217-9

1. Sinclair TKs (Computador) I. Beading, Bryan II. Título

84-0978

CDD — 001.64

Índice

1. O QUE OS COMPUTADORES PODEM FAZER, 17
(com sua ajuda)
2. LIGANDO O TK85, 21
3. O TK85 CONVERSA COM VOCÊ, 24
(usando o teclado)PRINT e NEW LINE
4. ERROS, 27
O que fazer quando o computador não entende RUBOUT.
5. O TK85 SE RECORDA, 29
Numerando as linhas de seu programa. RUN.
Espaços em branco
6. UM NOVO PROGRAMA, 33
Limando a memória do computador. NEW
7. STRINGS, 35
Aspas e caracteres
8. CURSORES, 37
Cursores K e L. Palavras-Chave
9. ESPAÇADORES, 39
Vírgula e Ponto-e-vírgula

10. O TK85 ESPERA POR SUA RESPOSTA, 41
INPUT

11. AS CAIXINHAS, 44
Variáveis string

12. OUTRO CAMINHO ATÉ AS CAIXINHAS, 47
Quando usar LET

13. NÚMEROS, 49
Adição, subtração, multiplicação e divisão.
Prioridades

14. MAIS CAIXAS, 52
Variáveis numéricas

15. FAZENDO ALTERAÇÕES, 54
O cursor de programa. EDIT.
As flechas. LIST

16. O TK85 DECIDE, 58
IF... THEN. CLS. STOP.
Ponto-e-vírgula

17. O TK85 REPETE O QUE FAZ, 62
GO TO. como sair de um loop.
BREAK

18. O TK85 ESCOLHE UM NÚMERO, 66
O cursor F. RND. INT

19. LOOPS PARA CONTAGEM, 73
Variáveis loop. FOR.. TO... NEXT.
PAUSE

20. DESENHANDO COM O TK85, 77
Graphics. Cursor G.
PRINT AT. CONTINUE

21. FIGURAS EM MOVIMENTO, 86
STEP

22. INKEY\$, 89

23. GUARDANDO O SEU PROGRAMA EM
FITA CASSETE, 92
SAVE. LOAD

24. PLANEJANDO SEUS PROGRAMAS, 98
Escrevendo o que objetiva fazer
Variáveis. Usando o REM

25. VOCÊ PODERIA FAZÊ-LO ASSIM, 103
Sugestões para programas propostos em capítulos
anteriores

26. JOGOS, 107

ÍNDICE ANALÍTICO, 118

Sobre a Autora

Carolyn Hughes entrou no mundo dos computadores quando sua família adquiriu um Sinclair ZX80. Mais tarde adquiriram também o ZX81, um ZX Spectrum e um micro BBC.

Devido a sua formação de educadora — já ensinou em várias escolas desde o primário até o secundário — Carolyn possui grande interesse no envolvimento de crianças com computadores, tanto no ensino como no lazer. Ela tem três filhos e dirige um clube de computadores no colégio onde dois de seus filhos estudam.

Tão bem quanto livros, ela também escreve softwares* educativos para o ZX81 e o ZX Spectrum.

* Softwares — Programas prontos para determinado trabalho.

Prefácio

Na época em que criei um clube de computador na escola em que meu filho estudava, usando o TK85 e o ZX Spectrum, pude observar o fascínio das crianças pelos computadores e o seu grande interesse em aprender a programá-los. No entanto, os manuais e livros de programação existentes no mercado foram escritos para adultos. Este livro tem como objetivo ser um guia, através dos conceitos do BASIC do TK85, do jovem leitor.

Nota sobre a Tradução

Alguns termos técnicos em processamento de dados têm uso consagrado no Brasil em inglês, por falta de um vocábulo em português que o traduza perfeitamente. Preferimos, nesta obra, conservá-los em inglês, para benefício do jovem que irá se acostumando, desde já, com o jargão dos técnicos em computação.

Tais termos são:

SOFTWARE	Pág. 12
STRING	Pág. 35
LOOP	Pág. 73

Introdução

Já existem vários livros sobre o TK85. Eles não foram escritos tendo você em mente. Este livro foi.

Embora você nunca venha a voar em um ônibus espacial para a lua, ou lutar com um dragão em uma caverna escura e subterrânea, o TK85 poderá lhe dar estas emoções.

Para imitar o voo do ônibus espacial, ou para criar um mundo de fantasias onde você derrota magos e bruxas, e encontra tesouros fabulosos, você precisa saber escrever um programa para o computador.

Programar um computador como o TK85 é muito divertido e extremamente fácil. Carolyn Hughes já ajudou muitas pessoas a aprenderem a programar o TK85, e neste livro optou por uma forma de fácil compreensão para abordar o assunto.

Você provavelmente não precisará da ajuda de ninguém para comandar o TK85, agora que possui este livro. Gaste apenas algumas horas com ele, com seu computador ligado, e logo estará fazendo seu computador obedecer a todos os seus comandos. Além disso, você verá que enquanto aprende coisas sobre o TK85, estará aprendendo sobre computadores em geral. Você logo saberá muito, bem mais que muitos adultos. E isto vale a pena.

Já é hora de começar a descobrir o quanto é divertido fazer um computador seguir os seus desejos.

Tim Hartnell,
Londres, Março de 1983

Tim Hartnell é o editor fundador da revista ZX COMPUTING, e autor de inúmeros livros populares sobre o assunto.

1

O que os Computadores Podem Fazer

Bem-vindo ao grupo dos usuários do TK85! Imagino que esteja curioso para saber o quanto esta pequena caixinha preta pode fazer. Bem, os computadores podem fazer muitas coisas, mas somente com nossa ajuda.



Apesar do TK85 parecer pequeno, a sua memória é ampla e pode executar coisas com extrema rapidez. Se você lhe der informações de alunos de uma escola, por exemplo, ele em poucos segundos pode lhe dar respostas tais como quantos meninos com menos de dez anos moram em determinada rua — trabalho este que seria muito lento e cansativo se fosse feito por você.

COMO ELE
É RÁPIDO!



JOÃO DA SILVA
EDUARDO COSTA
JAIME RIBEIRO
MARCUS VINICIUS
MARIO DOLACIO
FELIPE MARTINS

Ele também pode lhe ajudar nos estudos. Digamos que você tivesse de aprender os nomes e localizações dos rios da Austrália. O TK85 poderia mostrá-los na tela, perguntar o nome de cada um deles, e corrigir os que você errar.

NÃO, ESTE
É O RIO
VICTÓRIA



COMO ELE
É INTELIGENTE!

Computadores também podem executar tarefas nas fábricas. Muitos carros são feitos por robôs, conectados a computadores que trabalham com rapidez e precisão, além de não precisarem do intervalo para o cafezinho. No momento, os computadores que executam este tipo de tarefa são maiores e mais caros


do que seu TK85, mas você poderá descobrir que seu TK85 pode fazer algo semelhante, como controlar um modelo de ferrovia.



Até parece que nós, humanos, não somos mais necessários, não é mesmo? Mas na realidade somos, pois os computadores só fazem aquilo que lhes for ordenado, e nada podem fazer até que seres humanos lhes mostrem como.

Quando o seu TK85 foi feito, ensinaram-lhe a fazer cálculos matemáticos (adição, subtração, divisão, além de outros mais complicados).

$3974 \uparrow 7/143 + 29 = 1 \ 094609E+23$




Ensinaram-lhe também a entender uma linguagem chamada BASIC. Se quiser que o TK85 trabalhe para você, terá de aprender a usar o BASIC. O objetivo deste livro é dar início a esta aprendizagem. O BASIC é muito parecido com o inglês, mas você não deverá encontrar dificuldades.



2


Ligando o TK85

Coloque o seu TK85 em uma mesa, próximo a um aparelho de TV (colorida ou preto e branco).



O TK85 precisará ser conectado a uma tomada elétrica. Caso só se tenha uma, será necessário um adaptador, para que o televisor e o computador possam ser acoplados à mesma tomada.


Mais tarde você terá de usar também um gravador cassete. Logo, seria útil um adaptador com três ou mais tomadas.




Acompanhando o seu TK85, há um cabo como este:



O plug pontudo é conectado à saída de TV.



A outra ponta é conectada à entrada de VHF de seu televisor.




Agora você precisa da fonte de tensão (caixinha preta).




CONFIRA A
VOLTAGEM (110
OU 220) ANTES
DE LIGAR

Uma extremidade é conectada à tomada de corrente elétrica, e a outra ao seu TK85.



Abaixe totalmente o volume da TV. Ligue a fonte de tensão. Ligue a TV.

Coloque o seu aparelho de TV no canal 2 e ajuste a imagem até poder observar na tela isto:



O seu TK85 está pronto para trabalhar para você.

O TK85 usa a tela da TV para falar com você. No entanto, você tem que dizer-lhe o que falar.



Vamos fazer o TK85 dizer ALÔ para você. Você terá de digitar:

PRINT"ALO"

Vire a página para ver como se faz isto.



1. Aperte a tecla abaixo da palavra PRINT.
2. Aperte e segure a tecla SHIFT.
3. Aperte a tecla contendo as aspas " (é a mesma que usou para PRINT).
4. Solte a tecla SHIFT.
5. Digite ALO.
6. Aperte e segure a tecla SHIFT e aperte novamente ".

A tela de sua TV deve estar assim:



Agora aperte NEW LINE. Está logo abaixo da tecla que contém PRINT.

O TK85 obedeceu às suas ordens e escreveu ALO. A mensagem 0/0, na parte inferior da tela, é a forma usada pelo seu TK85 para lhe dizer que a ordem foi executada.

Aperte NEW LINE novamente e a tela estará limpa, exceto pelo K na parte inferior da tela.

4

Erros

Se você não falar corretamente a linguagem do computador, ele não vai lhe entender.

Tente digitar isto:

NÃO COLOQUE AS
SEGUNDAS "

PRINT"ALO



Aperte NEW LINE e veja o que acontece. Aparece um ponto de interrogação pois o TK85 não entendeu. Coloque agora as aspas " e aperte NEW LINE. Deve funcionar agora sem problemas.


O que fazer se você cometer um erro de grafia? Acima da tecla do canto superior direito está escrito RUBOUT. Esta tecla serve para apagar coisas na tela.

Digite:

PRINT"MACARAO"

Você esqueceu de um r!

Agora, aperte e segure a tecla SHIFT. Aperte a tecla RUBOUT e o cursor L apagará a letra que estiver à sua esquerda. Aperte novamente, e a próxima letra à esquerda do cursor será apagada!



Agora você pode digitar os seus comandos sem erros.


5

O TK85 se Recorda

O TK85 está obedecendo a seus comandos, mas será muito mais fácil para você se ele recordar suas instruções. Assim, você poderá fazê-lo repetir as mesmas coisas sem ter de digitá-las outra vez.

Ele se recordará se você colocar seus comandos em um programa.

Um programa é simplesmente uma lista de instruções, com um número para cada uma, de forma que o computador saiba em que ordem executá-las.



Costuma-se numerar as linhas de um programa de dez em dez, ao invés de uma em uma, para poder encaixar linhas possivelmente esquecidas.

Digite:

10 PRINT "LARANJAS"

NÃO CONFUNDA O NÚMERO 0
COM A LETRA O. O NÚMERO
TEM UMA BARRA QUE O
ATRAVESSA

Aperte NEW LINE.

A linha 10 se deslocará para o alto da tela.

Agora ela está na memória do computador.

Digite:

20 PRINT "BANANAS" (aperte NEW LINE)

30 PRINT "PERAS" (aperte NEW LINE)

Se após ter apertado NEW LINE a linha não for exatamente o que desejava, basta digitar novamente o que quiser, e apertar NEW LINE. O TK85 colocará a nova linha no lugar da anterior.



Quando seu programa estiver pronto, aperte RUN e em seguida NEW LINE para que o computador o execute.

Você deverá ter uma lista de frutas na tela de sua TV, e um 0 como mensagem na parte inferior da tela. O número 30 após 0/ significa que o TK85 concluiu as suas instruções na linha 30 do programa.


Aperte NEW LINE. Agora você pode ver a listagem de seu programa novamente.

Digite:

25 PRINT "UVAS" (aperte NEW LINE)

A linha 30 desceu, abrindo espaço para a 25. O TK85 conhece seus números!

Aperte RUN para rodar o programa de novo. Não esqueça de apertar NEW LINE após RUN.



Digite mais três linhas:

```
15 PRINT  
22 PRINT  
27 PRINT
```

NÃO SE ESQUEÇA DE APERTAR
NEW LINE APÓS CADA LINHA



Rode (RUN) o programa e veja o que estas linhas causaram.

6

Um Novo Programa

Você está pronto para escrever um novo programa? Diga isto ao computador apertando a tecla abaixo de NEW. Esta é a mesma tecla da letra A. Agora aperte NEW LINE.

A memória de seu TK85 está agora limpa outra vez para o seu próximo programa.

Façamos o computador imprimir o seu nome.

Digite:


```
10 PRINT"
```

Escreva seu primeiro nome. Aperte a tecla SPACE para dar um espaço em branco. Escreva seu último nome. Coloque as aspas " e aperte NEW LINE.



Rode (RUN) o programa e aperte NEW LINE para ver de novo a listagem de seu programa.

ANTES DE ESCREVER UM NOVO PROGRAMA, LIMPE SEMPRE A MEMÓRIA COM NEW E NEW LINE



EXERCÍCIOS:

Adicione algumas linhas ao seu programa para que o computador imprima (escreva) também o seu endereço. Procure pela vírgula e pelo comando de parada, STOP, no teclado do seu TK85. Eles estão em amarelo; logo, toda vez que tiver de usá-los, terá de pressionar simultaneamente a tecla SHIFT.

Se tiver problemas vá ao capítulo 25 na página 103.


7

Strings*

Até agora, todos os nossos programas tiveram strings.




Um string é uma seqüência de letras, números ou símbolos, tais como + ou *, chamados caracteres. C, \$, &, + são caracteres.



Quando você desejar imprimir um string na tela, basta colocar aspas " no começo e no fim do string. O computador imprime tudo que estiver entre as aspas. Você pode pensar nas aspas como nós nas pontas de um string para evitar que as letras se percam.

* Veja nota sobre tradução.


Existem também strings vazios, sem mesmo um espaço em branco entre as aspas. Tais strings são chamados nulos.



8

Cursorres

Imagino que você queira saber o que são os quadradinhos pretos com uma letra K ou L que freqüentemente aparecem na tela. Eles são chamados cursores, e indicam que o TK85 está pronto e esperando que você digite algo.



No topo de cada tecla, com exceção das superiores, existe uma palavra. Estas são chamadas de palavras-chave (em inglês Keywords). Se você apertar qualquer uma destas teclas quando o cursor for a letra K, a palavra-chave aparecerá na tela. Você já usou as palavras-chave PRINT e RUN.

Se você apertar uma tecla quando o cursor for a letra L, aparecerá na tela uma letra L.

A linha de teclas superior, que contém os números, imprimirá os números, estando o cursor tanto com a letra K como com L.


É MUITO FÁCIL LEMBRAR
K PARA "K"OMANDO E
L PARA LETRA



Aperte algumas teclas. Observe o cursor mudar de K para L. Quando tiver terminado, apague tudo usando a tecla RUBOUT. (Se precisar de ajuda volte ao capítulo 4.)

Em cada tecla, além das letras e palavras-chave, existem símbolos em amarelo. Para fazê-los aparecer na tela, pressione e segure a tecla SHIFT e aperte o símbolo que desejar.

A TECLA SHIFT EM AMARELO
PÁRA OS SÍMBOLOS E
PALAVRAS EM AMARELO



EXERCÍCIO:

Escreva um programa para imprimir uma linha de asteriscos * no topo da tela.

9

Espaçadores



Os espaçadores ajudam você a colocar as coisas onde quiser na tela.

Digite e rode (RUN) este programa:


```
10 PRINT "GUARDA"  
20 PRINT "CHUVA"
```



Agora troque a linha
10 digitando

```
10 PRINT "GUARDA";
```

Rode (RUN) o programa novamente.



A vírgula diz ao TK85 para mover até a outra metade da tela e então imprimir o próximo item. Digite

10 PRINT "GUARDA",



Você encontrará o ponto-e-vírgula (;) na mesma tecla da letra X. Como ele está em amarelo, é necessário pressionar a tecla SHIFT.

Quando rodar este programa (RUN) obterá:



O ponto-e-vírgula diz ao computador para não deixar qualquer espaço em branco antes de imprimir o próximo item.

Se desejar usar , ou ; como espaçadores, eles deverão estar fora das aspas. Se você os colocar dentro, farão parte do string e serão impressos na tela.



Agora aperte NEW e NEW LINE para um novo programa.

10

O TK85 Espera por sua Resposta

Agora você já sabe como fazer o TK85 dizer algo que você deseje. Você poderia fazê-lo perguntar algo.

Limpe a memória do computador apertando NEW e depois NEW LINE.


Agora digite este programa:

10 PRINT "QUAL E O SEU NOME?"

? ESTÁ EM AMARELO NA TECLA C.
VOCÊ PRECISA PRESSIONAR A
TECLA SHIFT COM ELA



Rode (RUN) o programa.



O TK85 não esperou por sua resposta!

Adicione estas duas linhas ao seu programa:

```
20 INPUT A$  
30 PRINT "ALÔ ";A$
```



Rode (RUN) novamente o programa.

Quando o cursor "L" aparecer na tela, digite seu nome e aperte NEW LINE.



Desta vez o TK85 não só esperou por sua resposta, mas lembrou-se do seu nome e dirigiu-se a você usando-o.

EXERCÍCIO:

Altere seu programa de forma que o TK85 deixe uma linha em branco entre perguntar o seu nome e dizer "ALÔ". Você poderá encontrar mais de uma maneira para ele fazer isto.

Eis como o TK85 lembra-se de seu nome.

Em sua memória existe uma parte especial só para guardar strings. Esta seção da memória é como 26 caixinhas, inicialmente vazias, com rótulos em branco.



Na linha 20 do seu último programa, o TK85 viu `INPUT A$`. Ele não pensou "Oba, vem algum dinheiro para mim!". Sabia que um string estava chegando e tinha de preparar uma caixa para guardá-lo.

Primeiro o TK85 verificou se alguma caixinha já tinha o rótulo `A$`. Não tinha. Então ele deu o nome `A$` a uma das caixas e esperou para ver o que você iria colocar nela.




Você digitou o seu nome, que então foi colocado dentro da caixa com o rótulo `A$`.



Na linha 30, depois de imprimir "ALO", o computador olhou dentro da caixa `A$` e imprimiu o string que lá encontrou.

O seu nome ainda está na caixa `A$`. Ele ficará lá até que você coloque outra coisa ou até apertar `RUN`, o que irá esvaziar todas as caixas.

Confira se seu nome ainda está lá, digitando `PRINT A$` e apertando `NEW LINE`.



`A$` é chamada de variável string, pois você pode variar, ou trocar o string que nela está guardado.

Você pode ter até 26 variáveis string num programa (lembra-se das 26 caixinhas?). Elas podem ser chamadas A\$, B\$ etc., até Z\$.

EXERCÍCIO:

Escreva um programa em que o TK85 lhe pergunte a cidade ou bairro onde você mora. Você digita a resposta e o computador deve imprimir o nome de sua cidade ou bairro, e em seguida deve dizer que é um lugar agradável de se viver. Certifique-se de que as palavras estão bem dispostas na tela.

12

Outro Caminho até as Caixinhas



Você pode dizer ao TK85 o que colocar em uma caixa de variável string antes de rodar o programa. Você o faz desta forma:

```
10 LET A$="COMPUTADOR"
```

Assim que você rodar este programa, o TK85 vai chamar uma caixa de A\$ e colocará a palavra "COMPUTADOR" dentro dela.



Isto será útil em um programa longo em que você queira que a palavra "COMPUTADOR" seja impressa várias vezes. Em vez de ter de digitar PRINT "COMPUTADOR" toda vez, você teria somente de digitar PRINT A\$, o que é mais rápido.

Lembre-se de não mandar o TK85 imprimir o conteúdo de uma caixa antes de colocar algo dentro dela. Ele não é mágico.



* 2/0 : O número 2 (dois) é um código de erro que significa variável não definida.

Outros códigos podem aparecer em caso de erro. Para identificá-los, consulte o manual que acompanha o TK85.

Chegamos agora em algo que o TK85 já sabe muito bem.

Vamos testar a sua matemática. Digite:
PRINT 132 + 499 (aperte NEW LINE)

Ele realmente é rápido em adição.


Tente agora uma questão de subtração:
PRINT 2954 - 877



Os sinais de multiplicação e divisão são diferentes dos usuais.



Ache-os no teclado. Eles estão em amarelo na linha inferior. Dê ao TK85 algumas questões de \times multiplicação e \div divisão. Ele pode operar com números bem grandes, mas quando forem muito grandes a resposta será dada de uma forma que você, provavelmente, não entenderá.




Se você lhe perguntar algo que envolva $+$ $-$ \times e \div , ele executará os cálculos em determinada sequência.

Peça-lhe: PRINT 2 + 3 * 4

Ele lhe deu a resposta que esperava, ou você achava que ele ia responder 20?

Em uma operação mista, o TK85 efetuará primeiro as multiplicações e divisões e depois as adições e subtrações.



Para que ele faça primeiro as adições e subtrações, estas expressões terão de ser colocadas entre parênteses.

Tente PRINT (2 + 3) * 4



EXERCÍCIOS:

1. Tente formular outras questões aritméticas e veja se os resultados correspondem ao esperado.
2. Descubra a diferença entre:

PRINT 5 - 3

e

PRINT "5 - 3"

No capítulo 10 você colocou um string dentro de uma caixa chamada A\$. Números também podem ser guardados em caixas.

Digite este programa:

```
10 INPUT A
20 INPUT B
30 LET C = A+B
40 PRINT C
```



Quando você rodar este programa, tudo que verá na tela será o cursor "L". Isto significa que o TK85 está esperando por um número, logo digite um qualquer e aperte NEW LINE. O computador colocará este número na caixa A (linha 10). O cursor "L" ainda está na tela, logo digite outro número e aperte NEW LINE. O computador soma os dois números e coloca a resposta dentro da caixa C (linha 30). Em seguida ele imprime o conteúdo da caixa C (linha 40).




As caixas numéricas (ou variáveis numéricas) podem receber letras do alfabeto como nomes ou palavras, se desejar. Você poderia ter uma variável numérica chamada TOTAL, ou RESPOSTA 5. Isto às vezes pode ser útil pelo fato de lembrar-lhe o que significa o número contido nesta caixa.

EXERCÍCIO:

Escreva um programa em que você ponha no computador dois números (chame-os de NUMERO 1 e NUMERO 2) e o computador os multiplique, colocando a resposta em RESPOSTA. Não esqueça de mandar o computador imprimir (PRINT) a resposta no final.

Até agora já ensinamos duas maneiras de mudar o que você digitou. Se não tiver apertado NEW LINE, você pode usar o RUBOUT para apagar. Se já tiver apertado NEW LINE, você pode trocar a linha do programa digitando-a novamente.

Você já deve ter notado  após o número de uma linha em seu programa — é chamado o cursor de programa e aponta a linha que acaba de ser introduzida no computador.

EU ESTOU NA
LINHA 40



```
10 INPUT NUMERO 1
20 INPUT NUMERO 2
30 LET RESPOSTA =
    NUMERO 1 * NUMERO 2
40 > PRINT RESPOSTA
```

Pressione (e segure) a tecla SHIFT e aperte a tecla EDIT (escrito em amarelo).



O TK85 copiará a linha e esta aparecerá na parte inferior da tela com o cursor logo após o número da linha.


Nas teclas 5, 6, 7 e 8 existem quatro flechas.



Com um dedo pressionando SHIFT, aperte 8.

O cursor se moverá para a direita.

Com um dedo pressionando SHIFT aperte 5 e o cursor se deslocará novamente, só que para a esquerda.



Utilizando as flechas, você poderá mover o cursor para a posição em que queira apagar ou inserir algo.

Se você tiver

40 PRINT RESPOSTA

na parte inferior da tela, mova o cursor até que esteja entre PRINT e RESPOSTA. Agora digite:

“A RESPOSTA E”;

Aperte NEW LINE e a sua nova linha 40, mais longa, estará no programa.

O que se deve fazer para alterar a linha 10, se necessário?

Com um dedo pressione SHIFT e aperte 7.

O cursor de programa subirá uma linha. Se você quiser que o cursor desça, pressione SHIFT e 6.



Quando o cursor estiver na linha correta, aperte SHIFT e EDIT e ele aparecerá na parte inferior da tela.

Pratique usando as quatro flechas e EDIT. Quando desejar que a linha na parte inferior da tela volte ao programa, basta apertar NEW LINE.

Uma forma rápida de mover o cursor de programa à linha desejada é usando LIST, que é a palavra-chave na tecla K. Aperte LIST e depois NEW LINE e o cursor se deslocará para a primeira linha do programa. LIST 20 levará o cursor para a linha 20, LIST 30 para a linha 30 e assim por diante.



Neste programa o TK85 mostra que sabe escolher as pessoas com quem fala.

```
10 PRINT "QUAL E O SEU NOME?"
20 INPUT A$
30 CLS
40 IF A$ < > "MICKEY MOUSE" THEN PRINT "EU
NAO CONVERSO COM PESSOAS CHAMADAS";A$
```



< > significa "diferente de".

Rode este programa.

Qual a função de CLS?




Na linha 40 o TK85 olhou dentro da caixa A\$. Se ele não encontrasse MICKEY MOUSE daria uma resposta grosseira.



Se ele encontrasse MICKEY MOUSE não escreveria nada. Vamos mudar isto.

Traga a linha 40 para a parte inferior da tela usando EDIT. (Se necessário reveja o último capítulo.)

Desloque o cursor para a direita até o fim da linha e adicione ;STOP




Aperte NEW LINE para devolver a nova linha 40 ao programa.

Digite agora:

50 PRINT "PRAZER EM CONHECE-LO"

Rode (RUN) o programa. Dê o seu nome como sendo MICKEY MOUSE e veja o que acontece.



Como MICKEY MOUSE estava em A\$, o TK85 pulou a linha 40 e foi direto para a linha 50.

Até agora nós só havíamos dado uma única instrução por linha para o computador. A linha 40 apresenta duas.

SE MICKEY MOUSE NÃO ESTIVER EM A\$,
PRIMEIRO EU PRECISO IMPRIMIR "EU
NÃO FALO COM PESSOAS CHAMADAS.....",
E DEPOIS PARAR (STOP).



É preciso ter um ponto-e-vírgula ; entre cada instrução.

EXERCÍCIOS:

1. Descubra o que acontece se você não colocar ;STOP na linha 40. EDIT a linha 40 e apague ;STOP. Em seguida rode o programa e dê o seu nome.

2. O prato favorito do TK85 é batatas fritas — escreva um programa em que:

- 10 O computador lhe pergunta qual o seu prato favorito.
- 20 Você digita a sua resposta que é colocada em F\$.
- 30 A tela é limpa.
- 40 Se você não responder "batatas fritas", o TK85 lhe diz que não gosta de seu prato e que ele prefere batatas fritas.
- 50 Se você responder "batatas fritas", o TK85 diz que também gosta.



O TK85 Repete o que Faz

Digite este pequeno programa.

```
10 INPUT A
20 INPUT B
30 GO TO 10
```



Rode este programa. Quando vir o cursor "L", digite um número. O TK85 o escreve e depois espera por outro. Como fazê-lo parar?



Em vez de colocar um número, pressione SHIFT e STOP, depois NEW LINE.



Tente agora:

```
10 INPUT A$
20 PRINT A$
30 GO TO 10
```


O cursor "L" indica que o computador está esperando por um string. Pressionar SHIFT e STOP desta vez não funciona. O TK85 vê STOP como um string e o coloca dentro de A\$.



Eis a forma de solucionar este problema. Apague as primeiras aspas do cursor, logo você tem L" no canto da tela. Pressione agora SHIFT e STOP. Você saiu do loop(laço).*

Quando estiver escrevendo um programa, você poderá acidentalmente entrar em um tipo de loop em que STOP não funcione. Eis aqui um exemplo:

```
10 LET N = 1
20 GO TO 10
```



Quando você rodar este programa, tudo que verá será uma tela em branco, sem cursor.

Você precisa sair do loop.

A tecla SPACE contém logo acima a palavra BREAK.

Aperte BREAK. Aperte NEW LINE. Aparecerá novamente a listagem de seu programa.

Loops podem ser muito úteis, se você tiver certeza de que o programa diz ao computador o momento em que ele deve sair do loop.

* Veja no fim desta obra Nota sobre tradução.

Eis um exemplo:

```
10 PRINT "ME DE UM NUMERO"
20 INPUT N
30 IF N = 0 THEN STOP
40 CLS
50 PRINT N;" * 12 = ";N*12
60 GO TO 10
```

Quando desejar parar este programa, basta digitar 0 (zero).



EXERCÍCIO:

Escreva um programa em que:

- 10 O computador pergunta por um nome.
- 20 Você digita um, e este é colocado em N\$.
- 30 Se você digitar "FIM" o computador pára (stop).
- 40 O computador imprime o nome.
- 50 O computador volta ao início do programa.

Em muitos jogos, é necessário que o computador escolha um número sem que você saiba qual. O TK85 tem em sua memória uma longa lista de números. Ele escolherá um, aleatoriamente*, quando usar RND.


RND está abaixo da tecla T. Para colocá-lo na tela você precisará da ajuda de outro cursor, o cursor F.



Após limpar a memória do computador com NEW, pressione SHIFT e NEW LINE simultaneamente.

Agora o cursor F deve estar na tela.

Aperte a tecla acima de RND.




* Aleatório = ao acaso, qualquer.

Apague RND e digite este programa:

```
10 PRINT RND
20 GO TO 10
```

Use sempre o cursor F para RND. Não digite as letras (R,N,D) uma a uma, senão o computador pensará que é uma variável numérica e procurará pela caixa rotulada RND.

Quando você rodar o programa, o TK85 imprimirá uma lista de números e então emitirá a mensagem 5/10, que significa que não há mais espaço na tela. Se desejar mais números aperte CONT (é a mesma tecla da letra C) e depois NEW LINE. Quando não quiser mais, aperte SPACE e depois NEW LINE.




Olhe para os números na tela. Eles não são o tipo de números nos quais gostaríamos de jogar, não? Seria impossível adivinhar qualquer um deles.




Nós precisamos de números maiores, sem todos esses decimais. Mude o programa para:

```
10 PRINT INT(RND * 10) + 1
20 GO TO 10
```



Rode o programa. Faça o computador encher algumas telas, e verá que todos os números que ele escolheu estão entre o 1 e 10.

Na linha 10, $RND * 10$ aumentou o número, `INT` cortou todos os dígitos após o ponto decimal*, e `+1` simplesmente serviu para que o computador nunca lhe desse o número 0.



Você sabe porque precisamos colocar $RND * 10$ entre parênteses? É para que o computador faça esta parte primeiro.

* O Ponto decimal corresponde, na notação em inglês, à vírgula que se usa no Brasil para separar a parte inteira de um número da fracionária. O TK85, como quase todos os computadores, usa a notação americana. Você terá que se acostumar com isso, o que não é difícil.

PRINT INT (RND * 10)



PRINT INT (RND * 10) + 1




Agora nós estamos prontos para um jogo de adivinhações.

```
10 LET A = INT(RND * 5) + 1
20 PRINT "ADIVINHE MEU NUMERO"
30 PRINT "VOCE ACHA QUE ELE E 1,2,3,4 ou 5?"
40 INPUT B
50 PRINT B
60 IF B = A THEN PRINT "CERTO. VOCE VENCEU."
70 IF B = A THEN STOP
80 PRINT "ERROU"
90 PRINT "TENTE OUTRA VEZ"
100 GO TO 40
```

QUAL NÚMERO
EU DEVO COLOCAR?

1 2 3 4 5



O TK85 jamais ganhará este jogo. Você tem quantas chances precisar. Façamos isto mais justo. Desta vez você só terá 3 chances.

Usaremos a caixa T para guardar o número de tentativas que você fizer. No começo você não terá nenhuma tentativa, portanto digite:

5 LET T = 0

Você faz sua tentativa de acertar na linha 40 (INPUT B). Logo após cada tentativa, precisamos adicionar 1 ao número em T.

45 LET T = T + 1

Quando você tiver feito 3 tentativas, não terá mais chances, logo precisaremos mudar as linhas 90 e 100.


```
90 IF T < 3 THEN PRINT "TENTE OUTRA VEZ"
100 IF T < 3 THEN STOP
```



< significa "menor que".

Deixe o TK85 congratular-se se conseguir vencê-lo!

120 PRINT "EU GANHEI. O NUMERO ERA";A



EXERCÍCIOS:


Escreva um programa em que:

- 10 O TK85 escolhe um número entre 1 e 100 e o coloca em A.
- 20 O TK85 escolhe um número entre 1 e 100 e o coloca em B.
- 30 Ele soma A com B e coloca o resultado em C.
- 40 Ele imprime A; '+'; B; '=';
- 50 Você digita a sua tentativa de resposta, que é colocada em D.
- 60 O TK85 imprime a sua resposta.
- 70 Se a sua resposta estiver correta, o TK85 deve congratulá-lo e parar o programa.
- 80 O TK85 diz a você a resposta correta.

19


Loops para Contagem

No último capítulo nós rotulamos uma caixa de variável numérica de T e a usamos para contar o número de tentativas que você efetuou.




O TK85 tem, na realidade, um conjunto de caixas especiais, chamadas variáveis loop, que são usadas só para contar. Quando você quiser usar uma, precisa dar-lhe um nome. Este nome deve ser uma letra do alfabeto. Eis um programa que irá imprimir o nome "TK85" vinte vezes.

```
10 FOR L = 1 TO 20
20 PRINT L;"TK85"
30 NEXT L
```



O TK85 começa rotulando uma caixa de variável loop de L e colocando 1 dentro dela.



Depois entra no loop.



Rode o programa, e verá o computador contando de 1 a 20 enquanto escreve seu próprio nome. Você pode fazê-lo contar mais devagar, dizendo-lhe para fazer pausas.

Digite

25 PAUSE 50




Quanto maior for o número que segue a palavra PAUSE, mais longa será a pausa.

Eis um programa que imprime a tabuada do 9 até 12X9:

```
10 FOR A = 1 TO 12
20 PRINT A;" * 9 =" ; A * 9
30 NEXT A
```

O computador fará o loop 12 vezes. Cada vez que ele atingir a linha 20, olhará dentro da caixa e escreverá o número que encontrar. Em seguida, escreverá o string " * 9 =". Depois disto ele calculará o valor de $A * 9$ e escreverá a resposta. NEXT A adiciona um (1) ao número contido em A e manda o computador voltar à linha 10 novamente.




EXERCÍCIO:

1. Diga ao computador para imprimir a tabuada de 16 até 16×20 .
2. Diga ao computador para imprimir 600 asteriscos, sem deixar qualquer espaço em branco entre eles.

20

Desenhando com o TK85

Em algumas teclas do teclado você verá os caracteres gráficos:



Você poderá usá-los para fazer figuras simples. Existe um cursor gráfico, o G, para imprimir estes caracteres.



Para colocá-lo na tela, pressione SHIFT e 9 (a qual tem a palavra graphics). Feito isto, ao pressionar qualquer tecla você verá que os caracteres normais aparecerão em forma inversa.

Se desejar usar os caracteres especiais, você terá de usar a tecla SHIFT.



Quando terminar de usar os caracteres gráficos, aperte NEW LINE e o cursor L voltará à tela.

Limpe a tela para fazer um pequeno programa que desenhará um carro.

```
10 PRINT "  █  "
20 PRINT "  □  "
```

Depois de ter digitado 10 print'', troque o cursor para G pressionando SHIFT e 9.


Os caracteres necessários são: █ □

Pressione SHIFT com QW e 6. Aperte NEW LINE para obter o cursor L. Agora você pode fechar as aspas ('').

A linha vinte (20) imprime as rodas, duas letras O com um espaço em branco entre elas.

Tente agora alguns de seus programas que desenhe figuras. Você pode tentar desenhar uma fábrica, um

cavalo, ou um ser extraterrestre. Ajudará se você desenhá-los em papel quadriculado, antes de escrever o programa.




Nem sempre se deseja a figura no topo esquerdo da tela.




Felizmente existe uma forma de dizer ao TK85 onde imprimir na tela o que se deseja.

Imagine que a tela de seu televisor seja dividida em quadrados como estes:





Existem 22 linhas, que são numeradas de 0 (linha superior) a 21 (linha inferior).

- 0
- 1
- 2
- 3
- 4
- 5
- 6
- 7
- 8 TK85
- 9
- 10
- 11
- 12
- 13
- 14
- 15
- 16
- 17
- 18
- 19
- 20
- 21




O MEU NOME
ESTÁ NA LINHA 8

Existem 32 colunas, que são numeradas de 0 (na esquerda) a 31 (na direita).



Para dizer ao TK85 onde imprimir um carácter, você deve primeiro lhe dar o número da linha, e depois o da coluna.



Digite:
PRINT AT 16,22;"X"



Tente imprimir outros caracteres em diferentes lugares. Lembre-se de que as linhas vão de 0 a 21 e as colunas de 0 a 31. Se você disser ao TK85 PRINT AT 30,40;"X", ele não poderá fazê-lo.






Para imprimir uma palavra em determinado ponto da tela, você deve dizer ao computador a linha e coluna em que deve estar a sua primeira letra. Tente colocar seu nome na tela, colocando a primeira letra na linha 11, coluna 13.

* Erro B = Número fora de faixa.



Vamos agora imprimir o carro na parte inferior da tela. O carro é composto de dois strings:

"" e "OO". Teremos as rodas na linha inferior, linha 21, e o corpo na linha 20. Vamos colocar ambos na coluna 6. Eis o programa:

```
10 PRINT AT 20,6; ""  
20 PRINT AT 21,6; ""
```



Façamos um disco-voador aterrissar num planeta. O disco começará no topo da tela (linha 0) e irá parar na linha 20, logo acima da superfície do planeta. Nós usaremos uma variável loop, chamada L, para contar as linhas à medida que o disco desce.

Eis o programa:

```
10 PRINT AT 21,0;" "
    "
20 FOR L = 0 TO 20
30 PRINT AT L,15;" "
40 PAUSE 20
50 IF L < 20 THEN PRINT AT L,15;" "
60 NEXT L
```



É CLARO QUE VOCÊ PODE
USAR SEU PRÓPRIO PLANETA E A
SUA NAVE ESPACIAL, EM VEZ
DA MINHA.




A linha 50 foi colocada para apagar o disco-voador antes que o próximo fosse impresso na linha abaixo. O meu disco-voador tem 6 caracteres de comprimento, logo eu preciso de seis espaços em branco para apagá-lo. Se o seu tiver sete caracteres, você precisará de sete espaços.

Por que preciso de : IF L < 20 THEN na linha 50?

PORQUE QUANDO O DISCO-VOADOR CHEGA
NA LINHA 20 DA TELA, ELE ATERRISSOU
E NÃO QUEREMOS APAGÁ-LO.



Para fazer o disco mover-se mais rápido, faça o número de pausa menor.




Você gostaria que o disco-voador decolasse novamente? Duas linhas do programa precisam ser alteradas.

```
20 for L = 20 TO 0 STEP -1  
50 PRINT AT L,15;"
```



Desta vez começaremos com o número 20 dentro da caixa L para que o disco-voador comece na linha 20 da tela. STEP -1 diz ao TK85 como contar. A cada passo ele precisa subtrair 1 (um) do número na caixa L.



EXERCÍCIOS:

1. Faça o disco-voador mover-se na horizontal.
2. Faça o disco-voador aterrissar e depois decolar.

22

INKEY\$

Em jogos de movimentos rápidos, você precisa que o computador corresponda no momento em que apertar uma tecla e não esperar até que você aperte NEW LINE.

Felizmente, existe uma palavra no BASIC que nos ajudará a resolver este problema. Esta palavra é INKEY\$, que significa "o carácter da tecla que está sendo pressionada" (KEY, em inglês, significa tecla).

Eis um programinha usando INKEY\$:

```
10 PRINT INKEY$;  
20 GO TO 10
```



Quando rodar este programa, inicialmente toda a tela ficará em branco. Aperte qualquer tecla. O computador imprimirá o carácter da tecla pressionada. Quanto maior o tempo que mantiver a tecla pressionada, maior será o número de vezes que o

carácter será impresso. Se você der um toque rápido na tecla, talvez consiga fazer o TK85 imprimir o carácter apenas uma única vez.

O programa é um pequeno loop. Na linha 10, o computador verifica se você está pressionando uma tecla. Se estiver, ele imprimirá o carácter da tecla. Se não estiver, imprimirá um espaço em branco e seguirá para a linha 20, e então voltará para a linha 10. De fato, o computador circula no loop muito rapidamente. Você entenderá porquê, se mantiver a tecla pressionada. O carácter é impresso várias vezes? INKEY\$ não é como INPUT. Quando o TK85 vê INPUT, ele espera que digite algo. INKEY\$ não espera, como verá no próximo programa.

```
10 PRINT "VAI ";  
20 IF INKEY$ = "P" THEN GO TO 40  
30 GO TO 10  
40 PRINT "STOP"
```



O loop vai da linha 10 a 30. O computador roda o loop, sem parar, até que você aperte "P". Apertar outra letra não adiantará. Tem de ser o P.

Troque a linha 30, colocando "Q" no lugar de "P", e veja a diferença que isto faz.

EXERCÍCIOS:

Escreva um programa para testar a sua rapidez. O computador diz a você para apertar P. Então ele inicia um loop do tipo "FOR. . .NEXT" contando de 1 a 1000. Se você apertar P, ele sairá do loop e dirá até que número contou.



Guardando o seu Programa em Fita Cassete

Quando você tiver digitado um programa razoavelmente longo, provavelmente desejará guardá-lo para rodá-lo de novo mais tarde. Se você desligar o seu TK85 o programa estará perdido. Mas antes de fazer isto, você poderá salvá-lo em uma fita cassete.

Você precisa de um gravador cassete e uma fita em branco. Um gravador portátil deve servir.



Conecte seu gravador à tomada (a não ser que funcione com pilhas). Introduza a fita.

Acompanha o seu TK85 um par de cabos como este.



Dois dos plugs devem ser de uma cor, e os outros dois de outra.

O TK85 tem dois soquetes na parte traseira, um ao lado do outro, chamados MIC e EAR.



Você verá que seu gravador também tem dois.

Conecte um dos plugs ao soquete MIC do TK85 e o outro (de mesma cor) ao soquete MIC do seu gravador. Coloque o outro plug (de outra cor) no soquete EAR do seu TK85, mas não conecte a outra extremidade ao gravador ainda.



Coloque a fita no começo e adiante um pouco, até que a fita marrom esteja aparecendo. Coloque o volume relativamente alto (em torno de 7, se tiver uma escala de 10 números). Se seu gravador tiver um controle de tonalidade, coloque o mais alto possível. Certifique-se de que a listagem de seu programa está na tela.



Você tem de dar um nome ao seu programa. Se estiver salvando o programa do capítulo 22 poderia chamá-lo de "VAI VAI". Você pode usar letras ou números como nome para seu programa. Este nome não poderá ter mais de 10 caracteres.

Aperte a tecla SAVE e digite o nome de seu programa entre aspas.



Coloque o gravador para funcionar e aperte NEW LINE. Você verá uma tela inicialmente apagada e depois várias linhas se movendo. A mensagem 0/0 diz que o TK85 terminou de gravar o programa. Pare o gravador cassete.

Agora vamos verificar se o programa foi gravado.



Rebobine a fita até o início do programa e ponha-o para "tocar".

Primeiramente você ouvirá um zumbido suave. Depois virão 5 segundos de silêncio, seguidos por um zumbido estridente e alto. Finalmente outro zumbido suave. Caso você não ouça nada disto, certifique-se de que o computador e o gravador estão ligados corretamente e repita o processo.

Com seu programa a salvo (gravado) na fita, você pode desligar seu TK85, ou escrever um novo programa.

Quando você quiser seu programa salvo de volta no computador, deve usar a palavra-chave LOAD (carregar).



Há duas formas de se dizer ao computador para LOAD (carregar).

Uma maneira é:

LOAD " "




que irá carregar o primeiro programa que o TK85 encontrar na fita.

A outra maneira é apertar LOAD e digitar o nome de seu programa entre aspas.



Se você tiver mais de um programa gravado em uma fita cassete, o segundo processo é melhor, pois o TK85 buscará o programa desejado.

Coloque a fita no começo. Conecte o último plug ao soquete EAR (ou AUX) do gravador e desconecte o plug do soquete MIC.



Aperte NEW LINE. Coloque o gravador para "tocar". Aparecerão na tela várias linhas e depois a mensagem 0/0.

Agora seu programa está pronto para rodar. Caso isto não ocorra, confira se as ligações do gravador e do computador estão corretas e repita o processo.


Quando se escreve um programa é melhor planejá-lo antes de digitá-lo, pois freqüentemente nos atrapalhamos ou esquecemos algo.



Vamos planejar um programa chamado "MEMORIA". Inicialmente escrevemos no papel tudo o que queremos que ele faça.

O computador irá escolher três números. Depois os mostrará na tela, um de cada vez. Ele lhe perguntará quais foram. Você digitará a resposta. O computador confere para saber se você está correto. Se estiver, escreverá duas vezes: "parabéns". Se você estiver errado, ele escreverá 10 vezes: "ERROU". Façamos agora uma lista de variáveis, ou caixas, que usaremos.


Os três números que o computador escolher serão colocados em N1, N2 e N3.



As suas respostas serão colocadas em T1, T2 e T3.



O loop que escreve "ERROU" será chamado A, e o que escreve "PARABÉNS" será chamado B.



Eis a lista completa:

Variáveis Numéricas:

N1..... primeiro número do computador
N2..... segundo " "
N3..... terceiro " "

T1..... sua primeira resposta
 T2..... " segunda "
 T3..... " terceira "

Variáveis Loop:

A..... loop que imprime "ERROU"
 B..... " " " " "PARABÉNS"

Quando se digita um programa, pode-se colocar linhas "lembretes". O computador não tomará conhecimento delas quando executar o programa, mas quando você olhar sua listagem mais tarde, lembrará o que cada parte faz.


A palavra usada pelo TK85 é REM (Abreviação de "remind" que, em inglês, significa lembrete).



Vamos então, agora, escrever o programa "MEMORIA", colocando lembretes:

```
10 REM * MEMORIA *
20 REM O COMPUTADOR ESCOLHE TRES
   NUMEROS
30 LET N1 = INT(RND * 10)
40 LET N2 = INT(RND * 10)
50 LET N3 = INT(RND * 10)
```

```
60 REM O COMPUTADOR MOSTRA OS NUMEROS
70 PRINT AT 10,15;N1
80 PAUSE 20
90 CLS
100 PRINT AT 10,15;N2
```



```
110 PAUSE 20
120 CLS
130 PRINT AT 10,15;N3
140 PAUSE 20
150 CLS
160 REM SUAS TENTATIVAS
170 PRINT "QUAL E O SEU PRIMEIRO NUMERO?"
180 INPUT T1
190 CLS
200 PRINT "QUAL E O SEU SEGUNDO NUMERO?"
210 INPUT T2
220 CLS
230 PRINT "QUAL E O SEU TERCEIRO NUMERO?"
240 INPUT T3
250 CLS
260 REM O COMPUTADOR CONFERE AS SUAS
   RESPOSTAS
270 IF T1 = N1 AND T2 = N2 AND T3 = N3 THEN
   GO TO 330
280 REM ROTINA PARA RESPOSTA ERRADA
```

AND ESTÁ NA TECLA 2



```
290 FOR A = 1 TO 10
300 PRINT "ERROU ERROU ERROU ERROU"
310 NEXT A
320 STOP
330 REM ROTINA PARA RESPOSTA CORRETA
340 FOR B = 1 TO 2
350 PRINT "PARABENS "
360 NEXT B
```



EXISTE UM
ESPAÇO
AQUI



DIVIRTA-SE!

25

Você Poderia Fazê-lo Assim

Aqui estão algumas maneiras de se fazer os exercícios propostos em capítulos anteriores. Existem várias maneiras de fazê-los. Estas são somente sugestões.

CAPÍTULO 6

```
10 PRINT "MARIO DOLACIO"
20 PRINT "RUA ARROIO PAVUNA,234"
30 PRINT "RIO DE JANEIRO"
```

Você se lembrou de colocar o que queria imprimir entre aspas?

CAPÍTULO 8

```
10 PRINT "*****"
```

Você precisa de 32 asteriscos para completar a linha superior da tela.

CAPÍTULO 10

```
10 PRINT "QUAL E O SEU NOME?"
```

```

20 INPUT A$
30 PRINT
40 PRINT "ALO";A$

```

ou

```

10 PRINT "QUAL E O SEU NOME?";
20 INPUT A$
30 PRINT "ALO";A$

```

Você encontrou alguma outra maneira?

CAPÍTULO 11

```

10 PRINT "AONDE VOCE MORA?"
20 INPUT P$
30 PRINT
40 PRINT P$
50 PRINT "E UM LUGAR AGRADAVEL DE SE VIVER"

```

CAPÍTULO 14

```

10 INPUT NUMERO1
20 INPUT NUMERO2
30 LET RESPOSTA = NUMERO1 * NUMERO2
40 PRINT RESPOSTA

```

CAPÍTULO 16

Exercício nº 2

```

10 PRINT "QUAL E O SEU PRATO PREDILETO?"

```

```

20 INPUT F$
30 CLS
40 IF F$ < > "BATATAS FRITAS" THEN PRINT "EU
   NAO GOSTO DISSO.PREFIRO BATATAS FRITAS."
50 IF F$ = "BATATAS FRITAS" THEN PRINT
   "CONCORDO COM VOCÊ.BATATAS FRITAS SAO
   UMA DELICIA."

```

CAPÍTULO 17

```

10 PRINT "DE-ME UM NOME"
20 INPUT N$
30 IF N$ = "FIM" THEN STOP
40 PRINT N$
50 GO TO 10

```

CAPÍTULO 18

```

10 LET A = INT(RND * 100) + 1
20 LET B = INT(RND * 100) + 1
30 LET C = A + B
40 PRINT A; " + ";B;" = ";
50 INPUT D
60 PRINT D
70 IF D = C THEN PRINT "PARABENS;"
80 IF D ≠ C THEN PRINT " VOCÊ ERROU, A
   RESPOSTA CERTA E:";C

```

Tente modificar o programa de forma que possa tentar outra vez, em caso de erro.

Depois modifique o programa de forma a ter somente três tentativas.

Olhe novamente o programa de adivinhações. Ele lhe dará algumas sugestões.

CAPÍTULO 19

Programa 1

```
10 FOR A = 1 TO 20
20 PRINT A;" * 16 = ";A * 16
30 NEXT A
```

Programa 2

```
10 FOR A = 1 TO 600
20 PRINT "A:";
30 NEXT A
```

ANTI-AÉREO

O objetivo deste jogo é destruir o maior número possível de naves com 50 tiros. É preciso levar em conta que, antes de os mísseis disparados alcançarem o topo da tela, não se pode voltar a disparar.

O número de pontos que se obtém em cada acerto está relacionado com a dificuldade de se acertar a nave. Assim, por exemplo, tem tanto mais valor acertar a nave quanto mais acima da tela ela tenha aparecido; da mesma forma, utilizar os projéteis dos canhões 1 e 2 proporciona maior número de pontos que os do 3 e, por sua vez, os do 1 mais que os do 2. Para dar os tiros, pressionam-se as teclas 1, 2 e 3, conforme o canhão que se deseje disparar.

Ao final do jogo, compara-se o número de pontos obtidos com o recorde estabelecido. Se o número for maior que o recorde anterior, o próprio programa o anunciará.

Lista de variáveis:

- A Linha de impressão da nave.
- B coluna de impressão do projétil.
- D Interruptor do projétil em curso.
- H Linha de impressão do projétil.

L coluna de impressão do projétil.
M Munição
P Pontos
R Recorde
T Retardo FOR para equalização de velocidades.

```

10 REM * ANTI-AEREO * ZX-81 E COMPATIVEIS
20 LET R=0
30 CLS
40 PRINT AT 21,0;" 1 "
50 PRINT AT 20,5;" 2 "; TAB 15;" 3 ";TAB
  25;" "
60 PRINT AT 0,1;"PONTOS: 0";" MUNIÇÃO: 50"
70 LET D=0
80 LET P=0
90 LET M=50
100 LET A=INT (RND*10)+2
110 FOR L=0 TO 28
120 PRINT AT A,L;" "
130 IF D THEN GO TO 210
140 IF INKEY$=" " THEN GO TO 300
150 IF INKEY$<"1" OR INKEY$>"3" THEN GO TO 300
160 LET B=10 * VAL INKEY$-4
170 LET D=1
180 LET M=M-1
190 LET H=19
200 PRINT AT 0,30;" ";AT 0,30; M
210 LET H=H-1
220 IF H<>A THEN GO TO 240
230 IF B>L AND B<L+4 THEN GO TO 350
240 PRINT AT H,B;"O";AT H+1,B;" "
250 IF H>2 THEN GO TO 320

```

```

260 LET D=0
270 PRINT AT H,B;" "
280 IF NOT M THEN GO TO 430
290 GO TO 320
300 FOR T=0 TO 1
310 NEXT T
320 NEXT L
330 PRINT AT A,29;" "
340 GO TO 100
350 PRINT AT H+1,B;" "
360 PRINT AT A,L;"BUUM"
370 PAUSE 1000
380 LET D=0
390 PRINT AT A,L;" "
400 LET P=P+(16-A) * (4-((B-6)/10+1))
410 PRINT AT 0,8;P
420 IF M THEN GO TO 100
430 IF P<=R THEN GO TO 470
440 PRINT AT 9,5;"ESTABELECIDO NOVO RECORD"
450 LET R=P
460 GO TO 480
470 PRINT AT 9,7;"O RECORD PERMANECE"
480 PRINT AT 11,16-(10+LEN STR$ R)/2;"DE ";R;"
  PONTOS"
490 PRINT AT 13,7;"QUER JOGAR DE NOVO ?"
500 IF INKEY$="S" THEN GO TO 30
510 IF INKEY$<>"N" THEN GO TO 500

```

FOTO MEMÓRIA

Eis um jogo desafiante que testará a sua capacidade de observação. O computador imprimirá dez letras,

aleatoriamente, na tela. Você terá um pequeno tempo para observá-las e em seguida serão apagadas. O computador imprimirá novamente as letras, só que desta vez com uma faltando. Qual será ela? Parece fácil, não é? Tente e verá!

```

10 REM FOTO MEMORIA
20 DIM A(20)
30 DIM B(10)
40 FOR R = 1 TO 20
50 LET A(R) = R + 37
60 NEXT R
70 FOR R = 1 TO 10
80 LET B(R) = A(INT(RND * 20) + 1)
90 NEXT R
100 CLS
110 FOR R = 1 TO 10
120 PRINT AT R,R * 2;CHR$ B(R)
130 NEXT R
140 LET W = INT(RND * 10) + 1
150 LET M = B(W)
160 LET B(W) = 0
170 FOR R = 1 TO 126
180 PLOT R/2,1
190 NEXT R
200 CLS
210 FOR R = 1 TO 10
220 PRINT AT R,R * 2;CHR$ B(R)
230 NEXT R
240 PRINT "QUAL A LETRA QUE ESTA FALTANDO?"
250 INPUT C$
260 IF CODE (C$) = M THEN GO TO 330

```

```

270 PRINT
280 PRINT
290 PRINT "ERRADO -- A LETRA QUE ESTA FALTANDO"
300 PRINT
310 PRINT "E A -";CHR$ M;"-"
320 GO TO 350
330 PRINT
340 PRINT "***** ACERTOU *****"
350 PRINT
360 PRINT "JOGA OUTRA VEZ? S/N "
370 INPUT Z$
380 IF Z$ = "S" THEN RUN
390 STOP

```

O BARQUEIRO

O programa propõe o seguinte problema lógico: Encomenda-se a um barqueiro levar de uma margem à outra três "passageiros": um lobo(L), uma cabra(C) e um fardo de palha ou outra forma de pasto(P). Na presença do barqueiro, os dois animais permanecem em calma, mas em sua ausência o lobo pode comer a cabra ou esta, o pasto; daí o sofrido barqueiro vê-se obrigado a pensar seriamente na ordem em que ele deve passar suas "mercadorias" de uma margem à outra, tendo em conta que seu barco só pode levar um "passageiro" por viagem.

Lista de variáveis:

- B\$ Barco representado na tela.
- C\$ Conteúdo do barco.
- S\$ Variável INPUT para "tirar do barco".

M\$ Variável INPUT para "pôr no barco".
O\$(2,3) Matriz de duas linhas e três colunas. A linha 1 representa o conteúdo da margem esquerda, e a 2 o da direita; cada uma das colunas representa os respectivos conteúdos das posições em cada uma das margens.
C Coluna de impressão do barco na sub-rotina de seu movimento.
P Número de passageiros.
I Variável auxiliar para FOR.

```

10 REM * O BARQUEIRO * SINCLAIR ZX-81 E
   COMPATIBLES
15 CLS
20 LET O=1
30 LET P=0
40 LET B$="███"
50 LET C$="███"
60 DIM O$(2,3)
70 LET O$(1)="LCP"
80 PRINT AT 2,0;"███ - O BARQUEIRO - ███"
90 PRINT AT 11,0;"███"
100 PRINT AT 12,0;"SSSSSSSSSSSSSSSSSSSSSSSSSSSSSS"
110 PRINT AT 10,0;O$(1)
120 PRINT AT 11,3;B$
130 GO SUB 160
140 GO SUB 280
150 GO TO 130
160 IF P=0 THEN GO TO 210
170 PRINT AT 21,0;"TIRAR DO BARCO?"
180 INPUT S$

```

```

190 GO SUB 350
200 IF O$(O)=" " THEN GO TO 240
210 PRINT AT 21,0;"POR NO BARCO?"
220 INPUT M$
230 GO SUB 470
240 RETURN
250 IF O$(2) ="LCP" THEN PRINT AT 21,0;"
   "; AT 5,9;"* MUITO BEM *"
260 IF O$(2) ="LCP" THEN STOP
270 RETURN
280 FOR I=3*(O-1)+22*(O-2) TO 22*(O-1)+3*(O-2)
   STEP 1-2*(O-2)
290 PRINT AT 11,I;B$
300 PRINT AT 10,I;C$
310 NEXT I
320 GO SUB 590
330 LET O=3-O
340 RETURN
350 IF S$=" " THEN RETURN
360 FOR I=3 TO 5
370 IF C$(I)=S$ THEN GO TO 400
380 NEXT I
390 IF I=6 THEN RETURN
400 LET C$(I)=" "
410 LET O$(O,I-2)=S$
420 PRINT AT 10,0*(O-1)+29*(O-2);O$(O)
430 PRINT AT 10,3*(O-1)+22*(O-2);C$
440 LET P=P1
450 GO SUB 250
460 RETURN
470 IF M$=" " THEN RETURN
480 FOR I=1 TO 3
490 IF O$(O,I)=M$ THEN GO TO 520

```

```

500 NEXT I
510 IF I=4 THEN RETURN
520 LET O$(0,I)=""
530 LET C$(I+2)=M$
540 PRINT AT 10,0*(O=1)+29*(O=2);O$(O)
550 PRINT AT 10,3*(O=1)+22*(O=2);C$
560 LET P=P+1
570 IF P=2 THEN GO TO 630
580 GO SUB 250
590 IF O$(O)="LC" OR O$(O)="CP" THEN GO TO 610
600 RETURN
610 PRINT AT 21,0;" ";AT 5,3;("O LOBO"
AND O$(O)="LC");("A CABRA" AND O$(O)=
"CP");" COMEU";("A CABRA" AND O$(O)="LC
");("O PASTO" AND O$(O)="CP")
620 GO TO 640
630 PRINT AT 21,0;" ";AT 5,3;"SINTO
MUITO, LOTADO."
640 PRINT
650 PRINT TAB 3;"QUER TENTAR OUTRA VEZ ?
(S/N)"
660 IF INKEY$="" THEN GO TO 660
670 IF INKEY$="S" THEN RUN
680 STOP

```

JOGO DAS LETRAS

Versão computadorizada de um passatempo antigo, este jogo consiste em ordenar um alfabeto, por meio do deslocamento das letras. Quando o jogo começa, aparece na tela uma matriz com as letras fora de ordem e com uma das posições em branco. Utilizando

as teclas correspondentes ao deslocamento do cursor (←, ↑, ↓, →), podemos mover a posição em branco dentro da moldura, conseguindo — com uma boa dose de lógica — a ordenação correta da matriz original.

Lista de variáveis:

- I Variável de diversas utilidades.
- J Variável de diversas utilidades.
- F Linha da matriz na qual está situado o espaço em branco.
- C Coluna da matriz na qual está situado o espaço em branco.
- FN Variável intermediária utilizada na atualização do valor F.
- CN Variável intermediária utilizada na atualização do valor C.
- X\$ É utilizado como armazenamento do valor da tecla pressionada e como temporário na atualização do conteúdo da matriz.
- M\$(5,5) Matriz de 5 linhas X 5 colunas, que contém as letras dispostas tal como aparecem na tela.

```

10 REM * JOGO DAS LETRAS *
20 FOR I=6 TO 16 STEP 2
30 PRINT AT I,10;" "
40 NEXT I
50 FOR I=7 TO 15 STEP 2
60 PRINT AT I,10;" ■ ■ ■ ■ ■ "
70 NEXT I
80 DIM M$(5,5)

```

```

90 FOR I=1 TO 5
100 FOR J=1 TO 5
110 LET M$(I,J)=CHR$(I+5*J+32-62*(I=5 AND J=5))
120 PRINT AT 2*I+5,2*J+9;M$(I,J)
130 NEXT J
140 NEXT I
150 LET F=5
160 LET FN=5
170 LET C=5
180 LET CN=5
190 FOR I=1 TO 200
200 LET X$=STR$(INT(RND*4)+5)
210 GO SUB 310
220 NEXT I
230 PRINT AT 20,1;"APERTE UMA TECLA PARA INICIAR"
240 PAUSE 1000
250 PRINT AT 20,1;"
"
260 GO SUB 280
270 GO TO 260
280 IF INKEY$ < > "" THEN GO TO 280
290 IF INKEY$ = " " THEN GO TO 290
300 LET X$=INKEY$
310 IF X$="5" AND C=1 THEN RETURN
320 IF X$="6" AND F=5 THEN RETURN
330 IF X$="7" AND FN=1 THEN RETURN
340 IF X$="8" AND CN=5 THEN RETURN
350 IF X$="5" THEN LET CN=CN-1
360 IF X$="6" THEN LET FN=FN-1
370 IF X$="7" THEN LET FN=FN-1
380 IF X$="8" THEN LET CN=CN+1
390 LET X$=M$(FN,CN)

```

```

400 LET M$(FN,CN)=M$(F,C)
410 LET M$(F,C)=X$
420 PRINT AT 2*FN+5,2*CN+9;M$(FN,CN)
430 PRINT AT 2*F+5,2*C+9;M$(F,C)
440 LET F=FN
450 LET C=CN
460 RETURN

```

Índice Analítico

AT	84
BREAK.....	64
CARACTERES.....	35
CLS	58
CURSOR F	66
CURSOR G	77
CURSOR K	37
CURSOR L	37
CURSOR DE PROGRAMA	54
EDIT	59
FOR	73
GO TO	62
GRAPHICS	77
IF	58
INKEY\$	89
INPUT	42
INT	69
LET	47
LIST	57
LOAD	95
NEW	33
NEW LINE	26
NEXT	73
PALAVRAS-CHAVE	37
PAUSE	75
PRINT	24

REM	100
RND	66
RUN	31
SAVE	94
STEP	88
STOP	59
STRING	35
THEN	58
TO	73
VARIÁVEL LOOP	73
VARIÁVEL NUMÉRICA	53
VARIÁVEL STRING	45
"	25
!	39
!	39
-	50
+	50
*	50
/	50
←	55
↓	56
↑	56
→	55
< >	58
<	71